
Abstract

In this report some numerical results are presented for a special
class of global optimization methods applied to molecular conforma-
tion problems. Following the results of a previously published method,
in this report the authors show how some modifications to the basic
structure of those algorithms enables us to obtain the optimal config-
uration for all clusters of Lennard-Jones atoms up to 80 atoms.

1 Problem definition

The problem we are facing is that of determining the configuration of mini-
mum potential energy of a cluster of identical atoms with no charge in three-
dimensional space. In this model all atoms are considered to be equal and
only pairwise interaction is included in the definition of the potential en-
ergy. Let N ≥ 2 be an integer representing the total number of atoms. The
Lennard-Jones (in short L-J) pairwise potential energy function is defined as
follows: if the distance between the centers of a pair of atoms is r, then their
contribution to the total energy is defined to be

v(r) =
1

r12
− 2

r6
(1)

and the L-J potential energy E of the cluster is defined as

E(X) = E(X1, . . . , XN) =
∑

i<j

v (‖Xi − Xj‖) (2)

where Xi ∈ R
3 represents the coordinates of the center of the i–th atom and

the norm used is the usual Euclidean one. An optimum L-J configuration
X� = {X�

1 , . . . , X
�
N} is defined as the solution of the global optimization

problem
LJN = E(X�) = min

X∈R3N
E(X). (3)

In a recent paper [2] the authors have shown how with the addition of a
penalty term in the potential energy all of the most difficult to find configura-
tions have been obtained, through a simple Multistart-like method. However
in that paper several cases which in the literature are considered to be easy,
could not be discovered. In this sense the two phase approach might seem to
be biased towards special classes of cluster conformations. We conjectured
this is not the case and this report has been produced in order to fill this
gap. It seems worthwhile to remind, however, that even if the approach
would have been deemed as biased, in any case this hypotetical biasedness
would have simply prevented the rediscovery of quite easy to find clusters,
most of which follow into the class of icosahedral ones.

We recall briefly here that the idea of two phase methods in cluster opti-
mization consists of using as a starting point for a local optimization of the
Lennard-Jones potential a cluster which is obtained as the result of a local
optimization, applied to a random configuration, of the following penalized
potential function: ∑

i<j

h (‖Xi − Xj‖) , (4)

where

h(r) =
1

r2p
− 2

rp
+ µr + β(max{0, r2 − D2})2, (5)

where p, µ, β,D ≥ 0.
Using a small set of parameters it was possible, in the above cited paper,

to obtain most of the putative optimum configurations for the Lennard-Jones
problem. The only cases which, in the range of clusters with up to 80 atoms,
the optimum could not be discovered with this approach and with the choice
we made for the parameters where the casesN = 62, 65, 66, 67, 68, 70, 71, 72, 73, 74, 78.
Also, in many cases in the range 61−−74, even if we could discover the op-
timum configuration, the rate of success was very low.

It is quite well known in the literature, however, that these are not con-
sidered as the most difficult to find configurations and it has been stated
in [1] that ”Any GO method ’worth its salt’ should be able to find all the
icosahedral minima and the truncated octahedron at N = 38. Success for the
other non-icosahedral minima would indicate that the method has particular
promise”. The same author, referring to our approach, says that methods

like our own are often capable of ”solving some ’hard’ instances’ while failing
for ’easier’ examples”.

In this paper we would like to show that using the same basic idea of
potential transformation alongside with more or less standard global opti-
mization methods we could succeed in discovering also the ”easy” cases.

Of course, in general, there is no interest in easy cases: but the fact
that modifications of the standard method are able of discovering even the
easy cases lets us assume that our class of methods is general enough to be
applicable to all the instances of Lennard-Jones, or other similar potential
minimization problems.

2 Two phase forward procedure

So called forward procedures are well known in the context of potential energy
minimization. They rely on the fact that in many cases, given an optimal
configuration for N atoms, the optimum configuration for N+1 has the same
geometric structure. Of course, even if this is true in some cases, the assertion
is false in particular for difficult configurations, which are indeed ”difficult”
because their structure impredictably changed when passing from N to N+1.
The idea of forward methods is the following: given an optimum configuration
of N atoms, a new atom is randomly placed in R

3; the resulting cluster of
N + 1 atoms is then optimized. Optimization, in the literature, is either
performed in two stages, first letting only the newly added atom vary and
then re-optimizing the whole cluster (3(N+1) variables). In our approach we
decided to eliminate the first stage and to consider, from the start, all of the
coordinates of the N +1 atoms as variables. However the local optimization
procedure adopted was the two-phase method, with penalty terms added. In
order to be able to obtain an optimum configuration it was observed that a
beneficial effect could be obtained by ”expanding” the N atom cluster prior
to optimization. This was performed by centering the cluster at its center
of mass and then multiplying each coordinate by a quantity greater than 1.
In the experiments we introduced also a deformation by using a multiplier
randomly generated between 1 and 1.3.

As the optimization in a forward method is generally quite fast, we de-
cided to use a standard local optimization method interfaced to AMPL to
perform our experiments.

The following is the AMPL source for the Lennard-Jones and the modified
potential function:

param semilato;

param xopt {1..n};

param yopt {1..n};

param zopt {1..n};

param p;

param lowx {1..n} <= 0,default -semilato;

param hix {1..n} >= 0, default semilato;

param lowy {1..n} <= 0, default -semilato;

param hiy {1..n} >= 0, default semilato;

param lowz {1..n} <= 0, default -semilato;

param hiz {1..n} >= 0, default semilato;

param maxiter;

param mu >0;

param beta;

param R;

param alpha >0;

var x {i in 1..n} >= lowx[i], <= hix[i];

var y {i in 1..n} >= lowy[i], <= hiy[i];

var z {i in 1..n} >= lowz[i], <= hiz[i];

var dist3{i in 1..n, j in 1..n: j < i};

var dist2{i in 1..n, j in 1..n: j < i};

var dist1{i in 1..n, j in 1..n: j < i};

minimize lennard:

sum {i in 1..n} sum {j in 1..n: j < i}

(1./dist3[i,j] - 2.)/dist3[i,j];

minimize lennard2:

sum {i in 1..n} sum {j in 1..n : j < i}

((1./dist2[i,j]^(p*0.5)-2.)/dist2[i,j]^(p*0.5)+mu*(dist1[i,j])

+ beta*max(0,dist2[i,j]-R*R)^2.);

s.t. defdist{i in 1..n, j in 1..n: j < i}:

dist3[i,j] =

((x[i] - x[j])^2. + (y[i] - y[j])^2. + (z[i] - z[j])^2.)^3.;

s.t. defdist1{i in 1..n, j in 1..n: j < i}:

dist1[i,j] =

sqrt((x[i] - x[j])^2. + (y[i] - y[j])^2. + (z[i] - z[j])^2.);

s.t. defdist2{i in 1..n, j in 1..n: j < i}:

dist2[i,j] =

(x[i] - x[j])^2. + (y[i] - y[j])^2. + (z[i] - z[j])^2.;

Here the objective lennard is the standard Lennard-Jones potential,
while lennard2 contains the definition of the modified potential. In order to
run the forward procedure, the following command file was used:

option display_precision 20;

option display_width 100;

option display1_col 0;

option substout 1;

model lj.mod;

option solver lancelot;

let semilato := 4;

let p := 6.;

let mu := 0.1 ;

let beta := 0;

let R := 0.;

param ottimoglobale;

let ottimoglobale := 0;

param nprove;

param xstart {1..n};

param ystart {1..n};

param zstart {1..n};

param barx;

param bary;

param barz;

param lowU;

param hiU;

let lowU := 1.0;

let hiU := 1.3;

data iniz.dat;

let barx := sum{i in 1..n-1} xstart[i]/(n-1);

let bary := sum{i in 1..n-1} ystart[i]/(n-1);

let barz := sum{i in 1..n-1} zstart[i]/(n-1);

let nprove := 0;

repeat while nprove < 20{

let nprove := nprove + 1;

display nprove;

let {i in 1..n-1} x[i] :=

(xstart[i]-barx)*Uniform(lowU,hiU);

let {i in 1..n-1} y[i] :=

(ystart[i]-bary)*Uniform(lowU,hiU);

let {i in 1..n-1} z[i] :=

(zstart[i]-barz)*Uniform(lowU,hiU);

let x[n] := Uniform(-semilato,semilato);

let y[n] := Uniform(-semilato,semilato);

let z[n] := Uniform(-semilato,semilato);

###############

# PHASE I #

###############

objective lennard2;

solve;

display lennard, lennard2;

###############

# PHASE II #

###############

objective lennard;

solve;

display lennard, lennard2;

printf "n %d nprove %d lennard %30.25lg p %lg mu

%lg esp %f %f beta %f D %f\n", n, nprove, lennard,p,mu,lowU,hiU,beta,R ;

if (ottimoglobale > lennard) then {

display x,y,z > opt.dat;

let {i in 1..n} xopt[i] := x[i];

let {i in 1..n} yopt[i] := y[i];

let {i in 1..n} zopt[i] := z[i];

let ottimoglobale := lennard;

}

display lennard, ottimoglobale;

}

let {i in 1..n} xstart[i] := xopt[i];

let {i in 1..n} ystart[i] := yopt[i];

let {i in 1..n} zstart[i] := zopt[i];

printf "param " > iniz.dat;

display xstart, ystart, zstart >> iniz.dat;

display ottimoglobale;

This procedure performs 20 runs each of which consists in using as the
starting configuration the coordinates contained in the file iniz.dat, adding
a single molecule, displacing each atom by a random amount from the barycen-
ter, and then optimizing using a two phase approach.

Using the modified potential function with µ = 0.1 and p = 6, with no
diameter penalization, we obtained the following success rate in 120 inde-
pendent tests (where the column ”succ” contains the number of runs which
led to to global optimum configuration in 120 experiments and % contains
the success rate):

N succ. %
61 4 3.33%
62 0 0.00%
63 8 6.67%
64 9 7.50%
65 0 0.00%
66 0 0.00%
67 3 2.50%
68 0 0.00%
69 0 0.00%
70 5 4.17%
71 3 2.50%
72 16 13.33%
73 0 0.00%
74 8 6.67%
75 0 0.00%
76 28 23.33%
77 2 1.67%
78 0 0.00%
79 4 3.33%
80 13 10.83%

It can be seen that some of the easy cases could be found with relatively
small computational effort using this forward procedure.

3 Correction procedure

It can be seen that some cases which the standard two phase method could
not discover are still lacking after running the forward method: we refer
in particular to N = 62, 65, 66, 68, 73, 78. For these cases, and others, we
tried with success the following correction procedure: after the two phase
optimization, a correction procedure was adopted with success. The following
AMPL commands report the core of the procedure:

fix x;

fix y;

fix z;

for {i in due} {

unfix x[i];

unfix y[i];

unfix z[i];

};

objective lennard1;

solve;

unfix x;

unfix y;

unfix z;

for {i in 1..n} {

let x[i] := x[i]*1.1;

let y[i] := y[i]*1.1;

let z[i] := z[i]*1.1;

};

objective lennard2;

solve;

objective lennard;

solve;

Here it is seen that a small subset of atoms is first considered as variable:
this subset actually consists of a random pair of atoms chosen at random
among the 20 atoms whose contribution to the total potential energy is max-
imum. Here the contribution of an atom to the potential is defined as

Ej(X) ==
∑

i�=j

v (‖Xi − Xj‖) (6)

and it is easily seen that

E(X) = 0.5
∑

j

Ej(X).

After randomly displacing two of the worst atoms, a phase I optimization,
with modified potential function, was carried out; from the optimum config-
uration found, which corresponds to a very easy local optimization in only
6 variables, all of the coordinates where considered as variables and, after
a slight expansion of 10% around the center of mass was imposed, again a
complete two phase optimization was performed.

With this correction procedure all of the lacking configurations where
successfully located in extremely short computational time

Conclusions

In this report we have shown how a standard two-phase method has been
modified in order to be able to obtain a set of tools which are capable of
discovering any known global optimum Lennard Jones cluster in the range
N ≤ 80. Indeed we extended our computations up to and found many
optimal configurations also for larger clusters.

As a conclusion it can be safely stated that, in order to discover globally
optimal Lennard-Jones clusters for a particular number N of atoms, the
following approach could be adopted:

1. First, if available, start from the best known cluster of N − 1 atoms
and apply the two-phase forward procedure;

2. then perform a few experiments with the correction method, randomly
displacing two of the worst atoms and reoptimizing; if the forward
procedure could not be used, the correction method could be applied
to some of the best clusters obtained in the following point 3:

3. perform a standard Multistart-like two phase optimization, both with
and without penalization terms on the diameter.

